begin integer Ntov,NClientov;

  real Tmod;

  integer j,U,i;

  outtext("Введите количество товаров: ");

  outimage;

  Ntov:=inint;

  outtext("Введите количество клиентов: ");

  outimage;

  NClientov:=inint;

SIMULATION begin

real array ShtrafNepost,ShtrafNepriem,StoimPost,StoimProd,THran,TPost,PZakaza,

           MinZakaz,MaxZakaz,VEdTov(1:Ntov),a(1:NClientov);

integer array TovTek,Norma(1:Ntov);

ref(head)array ZapasPartiyaTov(1:NTov);

real TotalShtraf,TotalPribyl,ShtrafNepostavka,ShtrafOtkaz,PoteriOtSpisaniya,VTovara,VSklada;

ref(Otpusk) Sclad;

  link class PartiyaTov(Kolvo,Date); integer Kolvo,Date;

    begin ref(Spisanie) SpisTovara;

    end PartiyaTov;

  process class Zapros(n); integer n; 

  comment Процесс - Клиент;

   begin integer array Zakaz(1:Ntov);

         integer i;

comment задержка между заказами;

    M: for i:=1 step 1 until Ntov do

    begin

      if draw(PZakaza(i),U) then

         Zakaz(i):=randint(2,7,U)

      else

         Zakaz(i):= 0

    end;

    Sclad.Client:-current;

    activate Sclad;

    hold(negexp(a(n),U));

    goto M;

   end Zapros;

  process class Otpusk(Client); ref(Zapros)Client; 

  comment Процесс - работа склада при отпуске товара;

  begin ref(PartiyaTov)X;

        integer i,Nabor;

  Nachalo: 

    for i:=1 step 1 until Ntov do

    begin

      if Client.Zakaz(i)>0 then

        begin 

          Nabor:=0; 

          for X:-ZapasPartiyaTov(i).first,X.suc while (Nabor<Client.Zakaz(i)and X=/=none) do

          begin

            Nabor:=Nabor+X.Kolvo; 

            cancel(X.SpisTovara);

comment отменили списание партии товара, так как её отпускают (быть может всю);

            X.out

          end;

          if Nabor ge Client.Zakaz(i) then

comment проверили, что товара набрали достаточно;

            begin

               TovTek(i):=TovTek(i) - Client.Zakaz(i);

               TotalPribyl:=TotalPribyl+Client.Zakaz(i)*StoimProd(i);

               if Nabor > Client.Zakaz(i) then

comment теперь проверили, нужно ли оставить на складе часть последней партии товара;

                 begin

                   activate X.SpisTovara at X.Date+THran(i);

                   X.Kolvo:=Nabor - Client.Zakaz(i);

                   VTovara:=VTovara-Client.Zakaz(i)*VEdTov(i);

                   if not(ZapasPartiyaTov(i).empty) then X.precede(ZapasPartiyaTov(i).first)

                   else X.into(ZapasPartiyaTov(i));

                 end;

            end

          else 

comment нужного количества товара не набрали;

            begin

              ShtrafNepostavka:=ShtrafNepostavka-(TovTek(i) - Client.Zakaz(i))*ShtrafNepost(i);

              VTovara:=VTovara-Nabor*VEdTov(i);

              TotalPribyl:=TotalPribyl+Nabor*StoimProd(i)

            end;

comment надо проверить необходимость пополнения запасов;

          if (TovTek(i) le Norma(i)) then activate new Zakaz(i) delay TPost(i);

        end;comment обработки ненулевого заказа;

    end; comment цикла по номенклатуре товаров; 

    passivate; goto Nachalo

  end Otpusk;

  process class Zakaz(Tovar); integer Tovar; comment Процесс - заказ товара;

  begin ref(PartiyaTov)X; 

        ref(Spisanie)S;

        integer VZakaza;

comment определим объём заказа;

    VZakaza:=randint(MinZakaz(i),MaxZakaz(i),U);

    if VZakaza-(VSklada-VTovara)>0 then

comment заказанному товару не хватает места на складе;

      begin 

        ShtrafOtkaz:=ShtrafOtkaz+(VZakaza-(VSklada-VTovara))*ShtrafNepriem(Tovar);

        VZakaza:=VSklada-VTovara;

      end;

      if VZakaza>0 then 

        begin

comment регистрируем партию товара и планируем её списание;

          X:-new PartiyaTov(VZakaza,time); X.into(ZapasPartiyaTov(Tovar));

          S:-new Spisanie(Tovar); X.SpisTovara:-S;

          activate S delay THran(Tovar);

        end;

  end Zakaz;

  process class Spisanie(Tovar); integer Tovar; comment Процесс - списание товара;

  begin ref(PartiyaTov)X; 

        integer VZakaza;

        integer i;

comment определим списываемую партию и её объём;

    X:-ZapasPartiyaTov(Tovar).first; X.out;

comment пересчитаем убытки от списывания;

    PoteriOtSpisaniya:=PoteriOtSpisaniya+X.Kolvo*StoimPost(Tovar);

  end Spisanie;

comment Главная программа;

  U:=1234783;

comment Ввод исходных данных;

  outimage;

  outtext("Введите объём склада: ");

  outimage;

  VSklada:=inreal;

  outimage;

  outtext("по каждому товару, в одну строку, через пробел, введите:");

  outimage;

  outtext("объём упаковки, время хранения, мин. и макс. размер заказа (в упаковках),");

  outimage;

  outtext("стоимости покупки и продажи, штрафы за непоставку и неприём");

  outimage;

  for i:=1 step 1 until NTov do

  begin

    VEdTov(i):=inreal;

    THran(i):=inreal;

    MinZakaz(i):=inreal;

    MaxZakaz(i):=inreal;

    StoimPost(i):=inreal;

    StoimProd(i):=inreal;

    ShtrafNepost(i):=inreal;

    ShtrafNepriem(i):=inreal;

  end;

  outimage;

  outtext("Введите интенсивности выставления заказов потребителями");

  outimage;

  for i:=1 step 1 until NClientov do

    a(i):=inreal;

  outimage;

  outtext("Введите распределение вероятностей заказа конкретных товаров потребителем");

  outimage;

  for i:=1 step 1 until NTov do

    PZakaza(i):=inreal;

  outimage;

  outimage;

  outtext("Введите для каждого товара начальный запас и критический уровень");

  outimage;

  for i:=1 step 1 until NTov do

  begin

    TovTek(i):=inint;

    Norma(i):=inint;

comment подсчитаем занятый объём склада;

    VTovara:=VTovara+TovTek(i)*VEdTov(i);

  end;

  outimage;

  outtext("Введите время моделирования");

  outimage;

  Tmod:=inreal;

comment создадим списки регистрации товаров;

  for i:=1 step 1 until NTov do

    ZapasPartiyaTov(i):-new head;

comment не забудем создать ссылку на склад (процесс Otpusk, по каждому клиенту);

  Sclad:-new Otpusk(none);

comment Моделирование начинается с первого заказа;

  for i:=1 step 1 until NClientov do

    activate new Zapros(i) at negexp(0.1,U);

  hold(TMod);

  TotalShtraf:=PoteriOtSpisaniya+ShtrafNepostavka+ShtrafOtkaz;

  outimage;

  outtext("================СТАТИСТИКА================");

  outimage;

  outtext("Потери составили:");

  outimage;

  outtext("От списания товара: ");

  outfix(PoteriOtSpisaniya,3,10);

  outimage;

  outtext("От отказов в поставке товаров: ");

  outfix(ShtrafNepostavka,3,10);

  outimage;

  outtext("От отказов в приёмке товаров: ");

  outfix(ShtrafOtkaz,3,10);

  outimage;

  outtext("Итого убытков: ");

  outfix(TotalShtraf,3,10);

  outimage;

  outtext("Прибыли составила: ");

  outfix(TotalPribyl,3,10);

  outimage;

  outtext("БАЛАНС: ");

  outfix(TotalPribyl-TotalShtraf,3,10);

  outimage;

 end SIMULATION;

end Main Program;

